

MICRO-1機械命令アセンブリ言語仕様

＜プログラム＞ ::= ＜プログラム頭部＞
 ＜プログラム本体＞
 ＜プログラム後部＞

＜プログラム頭部＞ ::= "TITLE" ＜タイトル名＞

＜プログラム本体＞ ::= ＜命令＞ { ＜命令＞ }

＜プログラム後部＞ ::= "END"

＜タイトル名＞ ::= ＜アルファベット＞ { ＜アルファベットまたは数字＞ }

＜命令＞ ::= ＜ラベル＞ ＜命令本体＞ | ＜命令本体＞

＜ラベル＞ ::= ＜名前＞

＜名前＞ ::= ＜アルファベット＞ { ＜アルファベットまたは数字＞ }

＜命令本体＞ ::= ＜命令グループ 1＞ | ＜命令グループ 2＞ | ＜命令グループ 3＞ |
 ＜命令グループ 4＞ | ＜命令グループ 5＞ | ＜命令グループ 6＞ |
 ＜命令グループ 7＞ | ＜命令グループ 8＞ | ＜命令グループ 9＞

＜命令グループ 1＞ ::= ＜op1＞ ＜rb＞ ", " ＜符号なし数＞ "(" ＜ra＞ ")" |
 ＜op1＞ ＜rb＞ ", " ＜符号なし数＞ |
 // " (ra)" が省略されたときは ra=0 とみなす
 ＜op1＞ ＜rb＞ ", "" "(" ＜ra＞ ")"

＜op1＞ ::= "ADD" | "SUB" | "AND" | "OR" | "XOR" | "MULT" | "DIV" | "CMP" | "EX"

＜命令グループ 2＞ ::= ＜op2＞ ＜rb＞ ", " ＜符号なし数＞

＜op2＞ ::= "LC" | "PUSH" | "POP"

＜命令グループ 3＞ ::= ＜op3＞ ＜rb＞ ", " ＜符号つき数＞

＜op3＞ ::= "SL" | "SA" | "SC" | "BIX"

＜命令グループ 4＞ ::= ＜op4＞ ＜rb＞ ", " ＜符号つき数＞ "(" ＜ra＞ ")" |
 ＜op4＞ ＜rb＞ ", "" "(" ＜ra＞ ")"
 // "符号つき数" が省略されたときは 0 とみなす

＜op4＞ ::= "LEA" | "LX" | "STX"

＜命令グループ 5＞ ::= ＜op5＞ ＜rb＞ ", " ＜アドレス＞

＜op5＞ ::= "L" | "ST" | "LA"

＜ra＞ ::= "0" | "1" | "2" | "3"

＜rb＞ ::= "0" | "1" | "2" | "3"

＜命令グループ 6＞ ::= ＜op6＞ ＜アドレス＞

＜op6＞ ::= "BDIS" | "BP" | "BZ" | "BM" | "BC" | "BNP" | "BNZ" | "BNM" | "BNC" | "B" | "BI" | "BSR"

＜命令グループ 7＞ ::= ＜op7＞ ＜装置＞

＜op7＞ ::= "RIO" | "WIO"

＜装置＞ ::= ＜装置名＞ | ＜装置番号＞

＜装置名＞ ::= "CR" | "LPT"
 // カードリーダー (装置番号=0)、ラインプリンタ (装置番号=1)

＜装置番号＞ ::= ＜10進数＞

```

<命令グループ 8> ::= <op8>

<op8> ::= "RET" | "NOP" | "HLT"

<命令グループ 9> ::= "DC" <定数> |
                    "DS" <10進数> |
                    "ORG" <開始アドレス>

<定数> ::= <符号つき数> | <文字定数> | <アドレス定数>

<文字定数> ::= "'" <文字> <文字>

<アドレス定数> ::= <名前>

<開始アドレス> ::= <16進数>

<アドレス> ::= <命令アドレス> | <命令アドレス> "+" <10進数> |
               <命令アドレス> "-" <10進数>

<命令アドレス> ::= "*" | <名前>
                // "*"はその命令自身のアドレスを表す

<符号なし数> ::= <10進数> | 'X' <16進数> | 'O' <8進数> | 'B' <2進数>

<符号つき数> ::= <符号> <符号なし数> | <符号なし数>

<符号> ::= "+" | "-"

<10進数> ::= <数字> { <数字> }

<16進数> ::= <16進数字> { <16進数字> }

<8進数> ::= <8進数字> { <8進数字> }

<2進数> ::= <2進数字> { <2進数字> }

<アルファベットまたは数字> ::= <アルファベット> | <数字>

<数字> ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"

<16進数字> ::= <数字> | "A" | "B" | "C" | "D" | "E" | "F"

<8進数字> ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7"

<2進数字> ::= "0" | "1"

<アルファベット> ::= "A" | "B" | "C" | "D" | "E" | ... | "X" | "Y" | "Z" |
                    "a" | "b" | "c" | "d" | "e" | ... | "x" | "y" | "z"

```

※ {}は0回以上の繰り返しを示す。

[eof]